DC Characterization of Semiconductor Power Devices

Product Note 4142B-1

Practical Applications Using the HP4142B Modular DC Source/Monitor



Table of Contents

1. Introduction

The HP 4142B Modular DC Source Monitor is a high speed, highly accurate, computer-controlled dc parametric measurement instrument for characterizing semiconductor devices. This product note uses an HP 4142B to show practical measurement examples that characterize semiconductor power devices.

Model number	Acronym	1-V range
HP 41420A Source Monitor Unit	HPSMU	40μV～200V、20fA～1A
HP 41421B Source Monitor Unit	MPSMU	40μV～100V、20fA～100mA
HP 41422A High Current Unit	HCU	40μV～10V、20μA～10A
HP 41423A High Voltage Unit	HVU	2 mV～1000V、2 pA～10mA
HP 41424A Voltage Source/Voltage Monitor Unit	VS／VMU	4 μV～40V
HP 41425A Analog Feedback Unit	AFU	

Table 1. The HP 4142B plug-in modules



You can mix and match different plug-in modules for unique application requirements.

Example configuration for measurements of devices on a wafer.

2. Application Examples

2.1 Automatic Extraction of Parameters

2.1.1. Automatic Measurements with a Module Selector

When you extract the dc parameters of a power device, you need to change the configuration for almost every parameter since each parameter requires a unique configuration of the instruments and measurement circuit. However, if the configuration can be changed automatically, the dc parameters can also be extracted automatically.

The HP 16087A Module Selector lets you change the configuration programmatically, thus freeing you from cumbersome configuration changes. This section shows a versatile example for automatically extracting the dc parameters of a MOSFET. The setups needed to extract each parameter are shown in Figure 1. The circuits in Figure 2 are functionally the same as in Figure 1, but electronically different. The setup in Figure 2 uses the module selector to automatically change the configuration.

An example of automatically extracting parameters by using the module selector is shown in Figure 3. The program listing of this example is shown in Figure 4.



Figure 1. Parameters for MOSFET and measurement circuits



Figure 2. You can easily change the connection of measurement modules with the module selector.



***** Parameter Measurement (MOS) ****

Vds(on)	=	5.02	(V)	(Id=2A, Vg=15V)	[HCU]
Rds(on)	=	2.51	(ohm)	(Id=2A, Vg=15V)	[HCU]
Vth	=	4.98	(V)	(Vd=10V)	[HCU]
Vth (by AFU)	=	3.512	(V)	(Vd=10V, Id=1mA)	[MPSMU]
yfs	=	.913	(S)		[HCU]
Igss	=	4.17E-11	(A)	(Vg=20V)	[MPSMU]
Bvdss	=	493.5	(V)	(Id=10mA)	[HVU]
Idss	=	.023216	(A)	(Vd=320V)	[HVU]

User 1 Caps Running

Continue

Figure 3. Sample measurement results for auto extraction of parameters.

```
10      OPTION BASE 1
20      COM /Meas/ @Hp4142,INTEGER Hcu,Hvu,Smu,Hpsmu
30      COM /Disp/ Vth,Vth_afu,Yfs,Igss,Bvdss,Idss,Vdson,Rdson
40      !
50      ASSIGN @Hp4142 TO 723
60      Hpsmu=2     ! slot 2
70      Smu=3       ! slot 3
80      Hcu=5       ! slot 5
90      Hvu=7       ! slot 7
100     !
110     Hcu_connect
120     Vds_on
130     Vth
140     Smu_connect
150     Igss
160     Vth_afu
170     Hvu_connect
180     Idss
190     Bvdss
200     Disp_res_mos
210     END
```

50-90 Initialization.
110-130 Connect HCU and measure Vds (on), Rds (on), Vth, yfs.
140-160 Connect SMU and measure Vth with AFU.
170-190 Connect HVU and measure Idss and BVdss.

Figure 4. Measurement program

Let's examine the benefits of using an HP 4142B to measure each parameter. For the ON state resistance measurement of a power MOSFET, a source of high current and a monitor for high resolution voltage are necessary. The HP 41422A High Current Source/Monitor Unit (HCU) can force a maximum current of 10A and can make high resolution measurements with a minimum voltage of 40μV. Therefore, the HCU can make precision measurements of the ON state resistance, which is an important parameter of power MOSFETs.

There are several ways to extract the threshold voltage (Vth) of a MOSFET. In this example, two methods are used. The first method measures the $\sqrt{Id}$-Vg characteristics, then draws a regression line and extracts as threshold voltage the X-axis value at the cross point of the regression line and the X-axis. The second method is much faster. An HP 41425A Analog Feedback Unit (AFU) and two HP 41421B Source/Monitor Units (SMUs) are connected in a feedback loop. The AFU monitors the output voltage of one SMU, which is connected to the gate of the MOSFET, and monitors the current of the other SMU, which is connected to the drain. When the drain current reaches a user-specified value, the voltage value of the gate (Vth) is extracted. Vth is usually measured by a combination of a High Power SMU (HPSMU) and a Medium Power SMU (MPSMU).

To measure the leakage current of a high power device, high voltage output and low current measurements are necessary. The HVU not only forces a maximum voltage of 1000V, but measures current with 2pA resolution.

For breakdown voltage measurements, the HVU has the quasi-pulse measurement mode[1] for precision measurements by minimizing the duration of the breakdown condition.

[1] Quasi-pulse measurement mode
The measurement sequence of this mode follows:

i) Force current specified by the user as current compliance.
ii) Monitor the voltage and calculate the voltage slew rate.
iii) When the Device Under Test (DUT) is in the breakdown condition, the current starts flowing rapidly and the voltage slew rate becomes flat. The unit detects this point, waits a user-specified delay time, and measures the output voltage.
iv) After the measurement, the output voltage is rapidly returned to the start voltage.

2.1.2. Enhancing Automatic Measurements by External Relay Control

You can open or short the output of the SMU by using the following methods:

OPEN Make the output current 0 A in current force mode.

SHORT Make the output voltage 0 V in voltage force mode.

For example, use these methods to open the base when you measure the BVceo of a bipolar transistor or to short the gate (grounded) when you measure the BVdss of a MOSFET, without ever having to remove the SMU from the base or the gate. When you measure certain parameters of a bipolar transistor or a MOSFET, the emitter of the bipolar transistor or the source of the MOSFET are usually connected to the ground unit (GNDU) and not to the SMU. Conversely, the connection between the GNDU and the device needs to be open when measuring other parameters, such as Icbo of a bipolar transistor. Opening and shorting the SMU make the configurations trouble-free.

This example shows how to programmatically measure the Icbo parameter of a power bipolar transistor by using an external relay. The example uses the Voltage Source (VS) of a Voltage Source/Voltage Meter Unit (VS/VMU) to control the external relay.

Before the measurement, make a measurement module as shown in Figure 5 by fixing the relay to the universal module (P/N 16088-60010). The default condition for the external relay is closed. By forcing a specified voltage to the relay from VS, the external relay is opened, and the connection between the GNDU and the emitter is opened. Figure 6 shows the measurement circuit, Figure 7 shows the measurement results,



Figure 6. Measurement circuit



Figure 7. Measurement result



Figure 5. Measurement module

and Figure 8 shows the program. An external relay used with a module selector (as shown in Figure 9) is an easy way to make even more versatile and automatic measurements. For instance, the connection to the GNDU and the transistor emitter can be opened to extract the Icbo parameter of a transistor.

```
10    OPTION BASE 1
20    ASSIGN @Hp4142 TO 723
30    Hpsmu=2
40    Hvu=7
50    Vs1=18
60    !
70    Vc=400               ! Vc = 400V
80    Iccomp=.01           ! Ic comp = 10mA
90    V_off=12             ! relay disconnect voltage
100   !
110   OUTPUT @Hp4142;"CN";Hvu,Hpsmu,Vs1
120   OUTPUT @Hp4142;"FMT";5
130   OUTPUT @Hp4142;"DV";Vs1,0,V_off
140   OUTPUT @Hp4142;"DV";Hpsmu,0,0,Iccomp
150   OUTPUT @Hp4142;"DV";Hvu,0,Vc,Iccomp
160   OUTPUT @Hp4142;"MM";1,Hvu
170   OUTPUT @Hp4142;"XE"
180   ENTER @Hp4142 USING "#,3A,12D,X";A$,Icbo
190   OUTPUT @Hp4142;"CL"
200   PRINT "Icbo = ";Icbo;"(A)"
210   END
```

20-90 Initialization.
110 Set the output switches of measurement modules to ON.
120 Specify format of the measurement data.
130 Open the relay OPEN by forcing 12 V to the relay from VS.
140 Ground the base.
160-200 Perform the measurement and display the results.

Figure 8. Measurement program



Figure 9. Auto extraction of parameters with external relay and module selector

2.2. Extending the Measurement Range

Since the HP 4142B can programmatically connect an HPSMU, HCU, or HVU to a device pin by using the module selector, you can make very wide-ranged measurements, as shown in the white area of Figure 10. In addition, you can use two HPSMUs, HCUs, or HVUs to extend the measurement range into the range indicated by the diagonal lines in Figure 10.

In this section, the measurement examples for devices that work in the extended voltage/current area of Figure 10 are shown.



Figure 10. Current and voltage range covered by the HP 4142B.

2.2.1. 2000V Measurement

One HVU can make breakdown tests of up to 1000V. You can increase the maximum voltage to 2000 V by using two HVUs in differential mode. The extended range is shown by diagonal lines in Figure 11. This is very useful for breakdown voltage measurements or current leakage measurements of 800/900V power transistors and SSRs (Solid State relays), both of which are used for switching power lines.

This example shows how to measure breakdown voltage of an 800 V power transistor. The measurement result, measurement circuit, and measurement program are shown in Figures 12-14.



Figure 11. Expanding the current and voltage range with two HVUs in series.

One HVU is connected to the collector and the other is connected to the emitter. First, −1000V (BV1) is applied to the emitter. Since the HVU is unipolar, you need to change the polarity of the HVU to negative in advance. Second, by using the break down command, a quasi-pulse is applied by the HVU connected to the collector. Then the voltage at the collector (BV2) is measured. By subtracting BV1 from BV2, you can get the actual breakdown voltage.



Figure 12. Measurement result



Figure 13. Measurement circuit

```
10    OPTION BASE 1
20    ASSIGN @Hp4142 TO 723
30    Hvu1=2             ! slot 2  ! hardware
40    Hvu2=4             ! slot 4  !    set up
50    !
60    Ic=5.0E-3          !(A)      !
70    Vccomp=1000        !(V)      ! parameters
80    Iccomp=-Ic         !(A)      !  initialization
90    Bv1=-1000          !(V)      !
100   !
110   OUTPUT @Hp4142;"CN";Hvu2,Hvu1
120   OUTPUT @Hp4142;"ERC1,2"
130   OUTPUT @Hp4142;"POL";Hvu1,1
140   OUTPUT @Hp4142;"FMT";5
150   OUTPUT @Hp4142;"MM";9,Hvu2
160   !
170   OUTPUT @Hp4142;"DV";Hvu1,0,Bv1,Iccomp*1.2
180   OUTPUT @Hp4142;"BDT";.2,2
190   OUTPUT @Hp4142;"BDM";1,0
200   OUTPUT @Hp4142;"BDV";Hvu2,0,400,1000,Iccomp
210   !
220   OUTPUT @Hp4142;"XE"
230   ENTER @Hp4142 USING "#,3A,12D,X";A$,Bv2
240   OUTPUT @Hp4142;"CL"
250   !
260   Bvceo=Bv2-Bv1
270   PRINT "PRE=",Bv1,"MEASURE=",Bv2,"STATUS=",A$
280   PRINT " BVceo =";Bvceo;"(V)"
290   !
300   END
```

10-90 Initialization.
120 Connect HVU by controlling the module selector.
130 Change the HVU output polarity.
140-150 Specify the format of the measurement data.
170-180 Output −1000V to the emitter and output quasi pulse from the collector.
200-250 Perform the measurement and display results.

Figure 14. Measurement program

2.2.2. 10A/20V Measurement

One HCU can output or measure up to 10A/10V. You can extend this range to 10A/20V by using two HCUs. The extended range is shown by diagonal lines in Figure 15. The extended measurement range makes it possible to evaluate devices that drive dc motors for cars.

This example shows how to measure Id-Vg characteristics by sweeping Vd from 0V to 20V. The measurement circuit, measurement result, and measurement program are shown in Figures 16-18.

One HVU is connected to the drain and the other is connected to the source, and an SMU is connected to gate. The measurement mode is set to dual pulse sweep measurement mode. The HCU is designed to output only pulse, so to perform a 0V to 20V sweep measurement, the sweep measurement is made two times: 0V to 10V and 10V to 20V.

In the first measurement, the HCU connected to the source forces 0V while the HCU connected to the drain forces sweep outputs varying from 0V to 10V. The Id parameter is measured in every step.

In the second measurement, each voltage value that was applied to the gate in the first measurement minus ten volts is applied to the gate. The HCU connected to the drain forces sweep outputs varying from 0V to 10V. This is equivalent to sweeping from 10V to 20V to the device. By sweeping Vd from 0V to 20V, these two measurements give the Id-Vd measurement as shown in Figure 17.



Figure 15. Expanding the current and measurement range with two HCUs in series.



Figure 16. Measurement circuit



Figure 17. Measurement result

```
10    OPTION BASE 1
20    DIM Vd(201),Id(5,201)
30    INTEGER Vd_no_step,Vg_no_step
40    !
50    ASSIGN @Hp4142 TO 723
60    Hcu1=4       !   Drain+
70    Hcu2=8       !   Drain-
80    Smu=6        !   Gate
90                 !   Source
100   Vs=-10                 ! (V)
110   Vd_start=0             ! (V)
120   Vd_stop=10             ! (V)
130   Vd_no_step=51          ! points
140   Id_comp=10             ! (A)
150   Vg_start=6             ! (V)
160   Vg_step=2              ! (V)
170   Vg_no_step=3           ! points
180   Ig_comp=.01            ! (A)
190   P_width=4.00E-4        ! (s)
200   !
210   Vd_step=(Vd_stop-Vd_start)/(Vd_no_step-1)
220   FOR Var1=1 TO Vd_no_step
230     Vd(Var1)=Vd_start+(Var1-1)*Vd_step
240   NEXT Var1
250   !
260   Lingraph(0,20,0,5,"VD (V)","ID (A)","ID-VD",0)
270   !
280   OUTPUT @Hp4142;"CN";Smu,Hcu2,Hcu1
290   OUTPUT @Hp4142;"ERC";1,3
300   OUTPUT @Hp4142;"FMT";5
310   !
320   OUTPUT @Hp4142;"PV";Hcu2,0,0,-.0002,-Idcomp
330   OUTPUT @Hp4142;"MM";3,Hcu2
340   OUTPUT @Hp4142;"XE"
350   OUTPUT @Hp4142;"DZ";Hcu2
360   OUTPUT @Hp4142;"BC"
370   !
380   FOR Var2=1 TO Vg_no_step
390     Vg=Vg_start+Vg_step*(Var2-1)
400     OUTPUT @Hp4142;"DV";Smu,12,Vg,.5
410     FOR Var1=1 TO Vd_no_step
420       OUTPUT @Hp4142;"PV";Hcu1,0,0,Vd(Var1),Id_comp
430       OUTPUT @Hp4142;"PDV";Hcu2,0,0,0,-Id_comp
440       OUTPUT @Hp4142;"PT";0,P_width,2.00E-1
450       OUTPUT @Hp4142;"MM";7,Hcu1
460       OUTPUT @Hp4142;"XE"
470       ENTER @Hp4142 USING "#,3A,12D,X";A$,Id(Var2,Var1)
480       PLOT Vd(Var1),Id(Var2,Var1)
490     NEXT Var1
500     OUTPUT @Hp4142;"DZ";Smu,Hcu1,Hcu2
510     PENUP
520   NEXT Var2
530   !
540   FOR Var2=1 TO Vg_no_step
550     Vg=Vs+Vg_start+Vg_step*(Var2-1)
560     OUTPUT @Hp4142;"DV";Smu,0,Vg,.5
570     FOR Var1=1 TO Vd_no_step
580       OUTPUT @Hp4142;"PV";Hcu1,0,0,Vd(Var1),Id_comp
590       OUTPUT @Hp4142;"PDV";Hcu2,12,0,Vs,-Id_comp
600       OUTPUT @Hp4142;"PT";0,P_width,2.00E-1
610       OUTPUT @Hp4142;"MM";7,Hcu1
620       OUTPUT @Hp4142;"XE"
630       ENTER @Hp4142 USING "#,3A,12D,X";A$,Id(Var2,Var1)
640       PLOT Vd(Var1)+10,Id(Var2,Var1)
650     NEXT Var1
660     PENUP
670   NEXT Var2
680   OUTPUT @Hp4142;"CL"
690   END
```

10-240	Initialization.
260	Draw graphic axis with HP 4142B Control software.
280-290	Set the output switches of measurement modules to ON.
320-360	Change the output polarity of the HCU connected to source, by outputting negative voltage.
380-520	Perform first sweep measurement (0-10V)
540-670	Perform second sweep measurement (0-20V)

Figuare 18. Measurement program

2.2.3. 20A/10V Measurement

The previous example shows a 10A/20V measurement by two HCUs in series. By using two HCUs in parallel, you can extend the measurement range up to 20A/10V. The measurement range extended by this configuration is shown by diagonal lines in Figure 19.

This example shows how to measure Ic-Vc characteristics of the power bipolar transistor. The Ic parameter can easily exceed 10A. The measurement circuit, measurement result, and measurement program are show in Figures 20-22.

The HCUs are connected in parallel between the collector and emitter as shown in Figure 20. The measurement mode is set to 2 channel pulsed sweep mode to synchronize the HCUs. The two HCUs are current sources that sweep current values from 0A to 10A. Current from the two HCUs flow into the bipolar transistor, which is equivalent to a sweep from 0A to 20A. By measuring the voltage at the top of either HCU, you can get Ic-Vc characteristics with 20A.



Figure 19. Expanding the current and measurement range with two HCUs in parallel.



Figure 20. Measurement circuit



Figure 21. Measurement result

```
10    OPTION BASE 1
20    DIM Ic(101),Vc(5,101)
30    INTEGER Vc_no_step,Ib_no_step
40    !
50    ASSIGN @Hp4142 TO 723
60    Hcu1=2       !   Drain
70    Hcu2=4       !
80    Smu=6        !   Gate
90                 !   Source
100   !
110   Ic_start=0              ! (A)
120   Ic_stop=9               ! (A)
130   Ic_no_step=101          ! points
140   Vc_comp=10              ! (A)
150   Ib_start=3.E-2          ! (A)
160   Ib_step=6.E-2           ! (A)
170   Ib_no_step=4            ! points
180   P_width=1.E-4           ! (s)
190   !
200   Ic_step=(Ic_stop-Ic_start)/(Ic_no_step-1)
210   FOR Var1=1 TO Ic_no_step
220     Ic(Var1)=Ic_start+(Var1-1)*Ic_step
230   NEXT Var1
240   !
250   Lingraph(0,10,0,20,"Vc (V)","Ic (A)","Ic-Vc",1)
260   !
270   OUTPUT @Hp4142;"CN";Hcu1,Hcu2,Smu
280   OUTPUT @Hp4142;"ERC";1,3
290   OUTPUT @Hp4142;"FMT";5
300   FOR Var2=1 TO Ib_no_step
310     Ib=Ib_start+Ib_step*(Var2-1)
320     OUTPUT @Hp4142;"DI";Smu,0,Ib,2
330     FOR Var1=1 TO Ic_no_step
340       OUTPUT @Hp4142;"PDI";Hcu1,0,0,Ic(Var1),Vc_comp
350       OUTPUT @Hp4142;"PI";Hcu2,0,0,Ic(Var1),Vc_comp
360       OUTPUT @Hp4142;"PT";.01,P_width,2.E-2
370       OUTPUT @Hp4142;"MM";7,Hcu1
380       OUTPUT @Hp4142;"XE"
390       ENTER @Hp4142 USING "#,3A,12D,X";A$,Vc(Var2,Var1)
400       PLOT Vc(Var2,Var1),2*Ic(Var1)
410     NEXT Var1
420     PENUP
430   NEXT Var2
440   !
450   OUTPUT @Hp4142;"CL"
460   END
```

10-230 Initialization.
250 Draw graphic axis with HP 4142B Control software.
270-280 Set the output switches of measurement modules to ON.
320 Force current to the base.
330-430 Perform sweep measurement by incrementing the current value which is forced by 2 channel pulse mode.

Figure 22. Measurement program

2.2.4. High Power Measurement (250mA × 100V, 125mA × 200V)

By connecting two HPSMUs in series or in parallel, you can make very high power measurements. This is effective for measuring the channel-on breakdown voltage of EL (Electro Luminescence) and PDP (Plasma Display Panel). The measurement range extended by this configuration is shown by diagonal lines in Figure 23.

This example shows how to measure Id-Vd characteristics in the high power measurement range by connecting two HCUs in parallel. The measurement circuit, measurement results, and measurement program are shown in Figure 24-26.

The white area inside the broken lines in Figure 25 shows the measurement range that can be covered with one HCU. Using two HPSMUs lets you extend the measurement range into the area indicated by diagonal lines.



Figure 23. Expanding the current and measurement range with two HPSMUs in parallel.



Figure 24. Measurement circuit



Figure 25. Measurement result

```
10    OPTION BASE 1
20    DIM Id(101),Vd(3,101)
30    INTEGER Id_no_step,Vg_no_step
40    !
50    ASSIGN @Hp4142 TO 723
60    Hpsmu1=2          !  Drain
70    Hpsmu2=7          !  Drain
80    Mpsmu=3           !  Gate
90    !
100   Id_start=0         ! (A)
110   Id_stop=.35        ! (A)
120   Id_no_step=101     ! points
130   Vd_comp=40         ! (V)
140   Vg_start=4         ! (V)
150   Vg_step=.1         ! (V)
160   Vg_no_step=3       ! points
170   Ig_comp=.1         ! (A)
180   !
190   Id_step=(Id_stop-Id_start)/(Id_no_step-1)
200   FOR Var1=1 TO Id_no_step
210     Id(Var1)=Id_start+(Var1-1)*Id_step
220   NEXT Var1
230   !
240   Lingraph(0,40,0,1,"Vd (V)","Id (A)","Id-Vd",0)
250   !
260   OUTPUT @Hp4142;"CN";Hpsmu1,Hpsmu2
270   OUTPUT @Hp4142;"FMT";5
280   FOR Var2=1 TO Vg_no_step
290     Vg=Vg_start+Vg_step*(Var2-1)
300     OUTPUT @Hp4142;"CN";Mpsmu
310     OUTPUT @Hp4142;"DV";Mpsmu,0,Vg,Ig_comp
320     FOR Var1=1 TO Id_no_step
330       OUTPUT @Hp4142;"DI";Hpsmu1,0,Id(Var1),Vd_comp
340       OUTPUT @Hp4142;"DI";Hpsmu2,0,Id(Var1),Vd_comp
350       OUTPUT @Hp4142;"MM";1,Hpsmu2
360       OUTPUT @Hp4142;"XE"
370       ENTER @Hp4142 USING "#,3A,12D,X";A$,Vd(Var2,Var1)
380       PLOT Vd(Var2,Var1),2*Id(Var1)
390     NEXT Var1
400     OUTPUT @Hp4142;"DZ";Mpsmu
410     PENUP
420   NEXT Var2
430   !
440   OUTPUT @Hp4142;"CL"
450   END
```

10-220 Initialization
240 Draw graphic axis with HP 4142B Control software.
280-310 Force specified voltage to the gate.
320-390 Perform sweep measurement by incrementing the current that is forced to the drain.

Figure 26. Measurement program

Appendix

Subprograms used in 2.1.1

```
230 Hcu_connect:SUB Hcu_connect
240     COM /Meas/ @Hp4142,INTEGER Hcu,Hvu,Smu,Hpsmu
250     OUTPUT @Hp4142;"CN";Hcu,Hpsmu
260     OUTPUT @Hp4142;"ERC";1,3
270   SUBEND
280   !
290 Smu_connect:SUB Smu_connect
300     COM /Meas/ @Hp4142,INTEGER Hcu,Hvu,Smu,Hpsmu
310     OUTPUT @Hp4142;"CN";Smu,Hpsmu
320     OUTPUT @Hp4142;"ERC";1,1
330   SUBEND
340   !
350 Hvu_connect:SUB Hvu_connect
360     COM /Meas/ @Hp4142,INTEGER Hcu,Hvu,Smu,Hpsmu
370     OUTPUT @Hp4142;"CN";Hvu,Hpsmu    !,Hpsmu
380     OUTPUT @Hp4142;"ERC";1,2
390   SUBEND
400   !
410 Vds_on:SUB Vds_on
420     COM /Meas/ @Hp4142,INTEGER Hcu,Hvu,Smu,Hpsmu
430     COM /Disp/ Vth,Vth_afu,Yfs,Igss,Bvdss,Idss,Vdson,Rdson
440     !
450     Vg=15            ! (V)
460     Igcomp=.01       ! (A)
470     Id=2             ! (A)
480     Vdcomp=10        ! (V)
490     !
500     OUTPUT @Hp4142;"FL";0,Hpsmu
510     OUTPUT @Hp4142;"FMT";5
520     OUTPUT @Hp4142;"PDI";Hcu,0,0,Id,Vdcomp
530     OUTPUT @Hp4142;"FL";0,Hpsmu
540     OUTPUT @Hp4142;"PT";0,5.0E-4,5.0E-2
550     OUTPUT @Hp4142;"PV";Hpsmu,0,0,Vg
560     OUTPUT @Hp4142;"MM";7,Hcu
570     OUTPUT @Hp4142;"XE"
580     OUTPUT @Hp4142;"DZ"
590     ENTER @Hp4142 USING "#,3A,12D,X";A$,Vdson
600     Rdson=Vdson/Id
610   SUBEND
620   !
630 Vth:SUB Vth
640     COM /Disp/ Vth,Vth_afu,Yfs,Igss,Bvdss,Idss,Vdson,Rdson
650     COM /Meas/ @Hp4142,INTEGER Hcu,Hvu,Smu,Hpsmu
660     DIM Vg(101),Sqrt_id(101),Id(101)
670     INTEGER Vd_no_step
680     !
690     Vd=10               ! (V)
700     Id_comp=5           ! (A)
710     Vg_start=0          ! (V)
720     Vg_stop=10          ! (V)
730     Vg_no_step=101      ! points
740     P_width=1.E-4       ! (s)
750     !
760     Vg_step=(Vg_stop-Vg_start)/(Vg_no_step-1)
770     FOR Var1=1 TO Vg_no_step
780       Vg(Var1)=Vg_start+(Var1-1)*Vg_step
790     NEXT Var1
800     !
810     OUTPUT @Hp4142;"FMT";5
820     OUTPUT @Hp4142;"PDV";Hcu,0,0,Vd,Id_comp
830     OUTPUT @Hp4142;"FL";0,Hpsmu
840     OUTPUT @Hp4142;"PT";0,P_width,2.E-2
850     OUTPUT @Hp4142;"MM";8,Hcu
860     OUTPUT @Hp4142;"PWV";Hpsmu,1,0,0,Vg_start,Vg_stop,Vg_no_step,.1
870     OUTPUT @Hp4142;"PT";0,P_width,1.E-2
880     OUTPUT @Hp4142;"XE"
890     OUTPUT @Hp4142;"DZ"
900     !
910     FOR Var1=1 TO Vg_no_step
920       ENTER @Hp4142 USING "#,3A,12D,X";A$,Id(Var1)
930       Sqrt_id(Var1)=SQRT(ABS(Id(Var1)))
940     NEXT Var1
950     !
960     N=45
970     Rline(N,Vg(*),Sqrt_id(*),A,B,K)
980     Vth=-A/B
990     Rline(N,Vg(*),Id(*),A,B,K)
1000    Yfs=1/B
1010  SUBEND
1020  !
1030 Igss:SUB Igss
```

230-270	Connect HCU to drain and HPSMU to gate.
290-330	Connect MPSMU to drain and HPSMU to gate.
350-390	Connect HVU to drain and HPSMU to gate.
410	Measures Vds (on) and Rds (on).
450-480	Define the parameters.
500-590	Measure Vds (on) by spot measurement.
600	Calculate Rds (on).
630	Measures Vth and yfs.
690-740	Define parameters.
760-790	Calculate Vg sweep voltages.
810-890	Measure Id-Vg characteristics by 2 channel pulsed sweep measurement.
910-930	Calculate $\sqrt{Id}$ values.
960-1000	Calculate Vth and yfs.
1030	Measures Igss.

```
1040     COM /Meas/ @Hp4142,INTEGER Hcu,Hvu,Smu,Hpsmu
1050     COM /Disp/ Vth,Vth_afu,Yfs,Igss,Bvdss,Idss,Vdson,Rdson
1060     !
1070     Vg=20               ! Vg+ (V)
1080     Ig_comp=1.E-4       ! 100(uA)
1090     Vd=0                !     (V)
1100     Idcomp=1.E-4        ! 100(uA)
1110     OUTPUT @Hp4142;"FMT";5
1120     OUTPUT @Hp4142;"DV";Hpsmu,12,Vg
1130     OUTPUT @Hp4142;"DV";Smu,0,0
1140     OUTPUT @Hp4142;"MM";1,Hpsmu
1150     OUTPUT @Hp4142;"XE"
1160     OUTPUT @Hp4142;"DZ"
1170     ENTER @Hp4142 USING "#,3A,12D,X";A$,Igss
1180   SUBEND
1190   !
1200 Vth_afu:SUB Vth_afu
1210     COM /Meas/ @Hp4142,INTEGER Hcu,Hvu,Smu,Hpsmu
1220     COM /Disp/ Vth,Vth_afu,Yfs,Igss,Bvdss,Idss,Vdson,Rdson
1230     !
1240     Vg_start=2         ! (V)
1250     Vg_stop=10         ! (V)
1260     Vg_rate=2000       ! (V/s)
1270     Igcomp=1.E-5       ! (A)
1280     Vd=10              ! (V)
1290     Id_target=.001     ! (A)
1300     Idcomp=.01         ! (A)
1310     Integ_time=4.5E-4  ! (s)
1320     Delay_time=1.E-4   ! (s)
1330     !
1340     OUTPUT @Hp4142;"ASV";Hpsmu,Vg_start,Vg_stop,Vg_rate,Igcomp
1350     OUTPUT @Hp4142;"AVI";Smu,Vd,Id_target,Idcomp
1360     OUTPUT @Hp4142;"ASM";1,1,Integ_time
1370     OUTPUT @Hp4142;"AT";0,Delay_time
1380     OUTPUT @Hp4142;"FMT";5
1390     OUTPUT @Hp4142;"MM";6
1400     OUTPUT @Hp4142;"XE"
1410     OUTPUT @Hp4142;"DZ"
1420     ENTER @Hp4142 USING "#,3A,12D,X";A$,Vth_afu
1430   SUBEND
1440   !
1450 Idss:SUB Idss
1460     COM /Meas/ @Hp4142,INTEGER Hcu,Hvu,Smu,Hpsmu
1470     COM /Disp/ Vth,Vth_afu,Yfs,Igss,Bvdss,Idss,Vdson,Rdson
1480     !
1490     Vd=320             !(V)
1500     Idcomp=1.0E-2      !(A)
1510     !
1520     OUTPUT @Hp4142;"FMT";5
1530     OUTPUT @Hp4142;"MM";1,Hvu
1540     OUTPUT @Hp4142;"DV";Hvu,0,0,Idcomp
1550     OUTPUT @Hp4142;"DV";Hpsmu,0,0,Idcomp
1560     WAIT 1.5
1570     OUTPUT @Hp4142;"DV";Hvu,0,Vd,Idcomp
1580     OUTPUT @Hp4142;"XE"
1590     ENTER @Hp4142 USING "#,3A,12D,X";A$,Idss
1600     OUTPUT @Hp4142;"DZ"
1610   SUBEND
1620   !
1630 Bvdss:SUB Bvdss
1640     COM /Meas/ @Hp4142,INTEGER Hcu,Hvu,Smu,Hpsmu
1650     COM /Disp/ Vth,Vth_afu,Yfs,Igss,Bvdss,Idss,Vdson,Rdson
1660     !
1670     Id=1.0E-2      !(A)
1680     Vdcomp=600     !(V)
1690     !
1700     OUTPUT @Hp4142;"FMT";5
1710     OUTPUT @Hp4142;"MM";1,Hvu
1720     OUTPUT @Hp4142;"DV";Hvu,0,0,1.E-2
1730     OUTPUT @Hp4142;"DV";Hpsmu,0,0,1.E-2
1740     OUTPUT @Hp4142;"DI";Hvu,0,Id,Vdcomp
1750     OUTPUT @Hp4142;"XE"
1760     ENTER @Hp4142 USING "#,3A,12D,X";A$,Bvdss
1770     OUTPUT @Hp4142;"CL"
1780   SUBEND
1790   !
1800 Disp_res_mos:SUB Disp_res_mos
1810     COM /Disp/ Vth,Vth_afu,Yfs,Igss,Bvdss,Idss,Vdson,Rdson
1820     COM /Lp_st/ INTEGER Lp_status,Loop_wait
1830     OUTPUT 2 USING "#,K";" K"
1840     PRINT "***** Parameter Measurement (MOS) ****"
1850     PRINT
```

1070-1100	Define parameters.
1110-1170	Measure Igss by spot measurement.
1200	Measures Vth with AFU.
1240-1320	Define parameters.
1340-1420	Measure Vth by analog feedback measurement.
1450	Measures Idss.
1490-1500	Define parameters.
1520-1600	Measure Idss by spot measurement.
1630	Measures BVdss.
1670-1680	Define parameters.
1700-1770	Measure BVdss by spot measurement.
1800-1950	Display measurement results.

```
1860    PRINT " Vds(on)       =",Vdson;"(V)     (Id=2A, Vg=15V)  [ HCU ]"
1870    PRINT " Rds(on)       =",Rdson;"(ohm)   (Id=2A, Vg=15V)  [ HCU ]"
1880    PRINT " Vth           =",DROUND(Vth,5);"(V)    (Vd=10V)  [ HCU ]"
1890    PRINT " Vth (by AFU) =",Vth_afu;"(V) (Vd=10V, Id=1mA)  [ MPSMU ]"
1900    PRINT " yfs           =",DROUND(Yfs,5);"(S)              [ HCU ]"
1910    PRINT " Igss          =",Igss;"(A) (Vg=20V)              [ MPSMU ]"
1920    PRINT " Bvdss         =",Bvdss;"(V)       (Id=10mA)      [ HVU ]"
1930    PRINT " Idss          =",Idss;"(A)    (Vd=320V)          [ HVU ]"
1940    PRINT
1950  SUBEND
1960  !
1970 Rline:SUB Rline(N,X1(*),Y1(*),A,B,K)
1980    OPTION BASE 1
1990    REAL X(5),Y(5)
2000    R2=0
2010    K=N
2020    WHILE R2<.9995 AND K<93
2030      X(1)=X1(K)
2040      X(2)=X1(K+2)
2050      X(3)=X1(K+4)
2060      X(4)=X1(K+6)
2070      Y(1)=Y1(K)
2080      Y(2)=Y1(K+2)
2090      Y(3)=Y1(K+4)
2100      Y(4)=Y1(K+6)
2110      Least(X(*),Y(*),A,B,R2)
2120      K=K+6
2130    END WHILE
2140  SUBEND
2150  !
2160 Least:SUB Least(X(*),Y(*),A,B,R2)
2170    OPTION BASE 1
2180    C=0             !  return value    !
2190    D=0             !     A :          !
2200    E=0             !     B : gradient !
2210    F=0             !     R2:          !
2220    G=0
2230    FOR I=1 TO 4
2240      C=C+X(I)
2250      D=D+Y(I)
2260      E=E+X(I)*X(I)
2270      F=F+Y(I)*Y(I)
2280      G=G+X(I)*Y(I)
2290    NEXT I
2300    A=(E*D-C*G)/(4*E-C*C)
2310    B=(4*G-C*D)/(4*E-C*C)
2320    R2=(A*D+B*G-D*D/4)/(F-D*D/4)
2330  SUBEND
```

1970-2140	Draw regression lines from every measurement point, find steepest line, and calculate line parameters.
2160-2300	Perform least square method and calculate parameters.

HEWLETT PACKARD

For more information, call your local HP sales office listed in your telephone directory or an HP regional office listed below for the location of your nearest sales office.

United States:
Hewlett-Packard Company
4 Choke Cherry Road
Rockville, MD 20850
(301) 670-4300

Hewlett-Packard Company
5201 Tollview Drive
Rolling Meadows, IL 60008
(708) 255-9800

Hewlett-Packard Company
5161 Lankershim Blvd.
No. Hollywood, CA 91601
(818) 505-5600

Hewlett-Packard Company
2015 South Park Place
Atlanta, GA 30339
(404) 955-1500

Canada:
Hewlett-Packard Ltd.
6877 Goreway Drive
Mississauga, Ontario L4V 1M8
(416) 678-9430

Japan:
Yokogawa-Hewlett-Packard Ltd.
15-7, Nishi Shinjuki 4 Chome
Shinjuki-ku
Tokyo 160, Japan
(03) 5371-1351

Latin America:
Hewlett-Packard
Latin America Region Headquarters
Monte Pelvoux No. 111
Lomas de Chapultepec
11000 Mexico, D.F. Mexico
(525) 202-0155

Australia/New Zealand:
Hewlett-Packard Australia Ltd.
31-41 Joseph Street
Blackburn, Victoria 3130
Melbourne, Australia
(03) 895-2895

Far East:
Hewlett-Packard Asia Ltd.
22/F Bond Centre
West Tower
89 Queensway
Central, Hong Kong
8487777

European Headquarters
Hewlett-Packard S.A.
150, Route du Nant d'Avril
1217 Meyrin 2
Geneva-Switzerland
41/22 780 8111

In Europe, please call your local HP sales office or representative:

Austria:
(0222) 2500-0

East Central Europe, USSR, and Yugoslavia:
Vienna - Austria
(0222) 2500-0

Belgium and Luxembourg:
(02) 761 31 11

Denmark:
(42) 81 66 40

Finland:
(0) 88 721

France:
(1) 69 82 60 60

Germany:
(06172) 16 0

Greece:
(01) 68 28 811

Iceland:
(91) 67 10 00

Ireland:
(01) 88 33 99

Italy:
(02) 92 19 91

Netherlands:
(020) 547 6669

Norway:
(02) 24 60 90

Spain:
900 123 123

Sweden:
(08) 750 20 00

Switzerland:
(057) 31 21 11 (Head office)
(022) 780 41 11 (Suisse Romande)
(046) 05 15 05 (Customer Information Center)

South Africa:
HiPerformance Systems
(011) 802 5111

Turkey:
175 29 70

U.K.:
(0344) 369 369

Middle East and Africa:
Geneva-Switzerland
41/22 780 7111

Data subject to change
Printed in U.S.A. 9/91
5091-2744E